Chapter 8 Gram-Schmidt Orthogonalization (September 8, 2010)

Jørgen Pedersen Gram (June 27, 1850 – April 29, 1916) was a Danish actuary and mathematician who was born in Nustrup, Duchy of Schleswig, Denmark and died in Copenhagen, Denmark. Important papers of his include *On series expansions determined by the methods of least squares*, and *Investigations of the number of primes less than a given number*. The process that bears his name, the Gram–Schmidt process, was first published in the former paper, in 1883.



http://en.wikipedia.org/wiki/J%C3%B8rgen_Pedersen_Gram

Erhard Schmidt (January 13, 1876 – December 6, 1959) was a German mathematician whose work significantly influenced the direction of mathematics in the twentieth century. He was born in Tartu, Governorate of Livonia (now Estonia). His advisor was David Hilbert and he was awarded his doctorate from Georg-August University of Göttingen in 1905. His doctoral dissertation was entitled *Entwickelung willkürlicher Funktionen nach Systemen vorgeschriebener* and was a work on integral equations. Together with David Hilbert he made important contributions to functional analysis.

http://en.wikipedia.org/wiki/File:Erhard_Schmidt.jpg

8.1 Gram-Schmidt Procedure I

Gram-Schmidt orthogonalization is a method that takes a non-orthogonal set of linearly independent function and literally constructs an orthogonal set over an arbitrary interval and with respect to an arbitrary weighting function. Here for convenience, all functions are assumed to be real.

$u_n(x)$ linearly independent non-orthogonal un-normalized functions

Here we use the following notations.

$u_n(x) = x^n$ (n = 0, 1, 2, 3, …..).

$\psi_n(x)$ linearly independent orthogonal un-normalized functions

$\varphi_n(x)$ linearly independent orthogonal normalized functions

with

$$\int_a^b \varphi_i(x)\varphi_j(x)w(x)dx = \delta_{i,j} .$$

Starting with $n = 0$, let

$$\psi_0(x) = u_0(x),$$

$$\varphi_0(x) = \frac{\psi_0(x)}{\sqrt{\int_a^b \psi_0{}^2 w(x)dx}}.$$

For $n = 1$,

$$\psi_1(x) = u_1(x) + a_{10}\varphi_0(x).$$

We demand that $\psi_1(x)$ be orthogonal to $\varphi_0(x)$.

$$\int_a^b \psi_1(x)\varphi_0(x)w(x)dx = 0 = \int_a^b u_1(x)\varphi_0(x)w(x)dx + a_{10}\int_a^b \varphi_0(x)^2 w(x)dx$$

or

$$a_{10} = -\int_a^b u_1(x)\varphi_0(x)w(x)dx.$$

Normalizing, we define

$$\varphi_1(x) = \frac{\psi_1(x)}{\sqrt{\int_a^b \psi_1{}^2 w(x)dx}}.$$

Generalizing, we have

$$\varphi_i(x) = \frac{\psi_i(x)}{\sqrt{\int_a^b \psi_i{}^2 w(x)dx}},$$

where

$$\psi_i(x) = u_i(x) + a_{i0}\varphi_0(x) + a_{i1}\varphi_1(x) + a_{i2}\varphi_2(x) + ... + a_{i,i-1}\varphi_{i-1}(x).$$

The coefficient (a_{ij}) are given by

$$a_{ij} = -\int_a^b u_i(x)\varphi_j(x)w(x)dx .$$

since

$$\int_a^b \psi_i(x)\varphi_j(x)w(x)dx = 0$$

$$= \int_a^b \{u_i(x) + a_{i0}\varphi_0(x) + a_{i1}\varphi_1(x) + a_{i2}\varphi_2(x) + ... + a_{i,i-1}\varphi_{i-1}(x)\}\varphi_j(x)w(x)dx$$

$$0 = \int_a^b u_i(x)\varphi_j(x)w(x)dx + a_{ij}\int_a^b [\varphi_j(x)]^2 w(x)dx .$$

8.2 Example: Hermite polynomials

The Hermite polynomial $H_n(x)$ can be derived from the orthogonalization,

$$u_n(x) = x^n \quad (n = 0,1,2,3, \quad),$$

with the weight function

$$w(x) = e^{-x^2} .$$

Starting with $n = 0$, let

$$\varphi_0(x) = \frac{u_0(x)}{\sqrt{\int_{-\infty}^{\infty} e^{-x^2} dx}} = \frac{1}{\pi^{1/4}}$$

For $n = 1$, let

$$\psi_1(x) = u_1(x) + a_{10}\varphi_0(x) ,$$

$$0 = \int_{-\infty}^{\infty} \varphi_0(x)e^{-x^2}\psi_1(x)dx = \int_{-\infty}^{\infty} \varphi_0(x)e^{-x^2}[u_1(x) + a_{10}\varphi_0(x)]dx$$

or

$$0 = \int_{-\infty}^{\infty} \varphi_0(x)e^{-x^2}u_1(x)dx + a_{10}$$

or

$$a_{10} = -\int_{-\infty}^{\infty} \varphi_0(x) e^{-x^2} u_1(x) dx = -\frac{1}{\pi^{1/4}} \int_{-\infty}^{\infty} x e^{-x^2} dx = 0$$

Then we have

$$\psi_1(x) = u_1(x) .$$

$$\varphi_1(x) = \frac{u_1(x)}{\sqrt{\int_{-\infty}^{\infty} x^2 e^{-x^2} dx}} = \frac{\sqrt{2}x}{\pi^{1/4}} .$$

For $n = 2$, let

$$\psi_2(x) = u_2(x) + a_{20}\varphi_0(x) + a_{21}\varphi_1(x) ,$$

$$0 = \int_{-\infty}^{\infty} \varphi_0(x) e^{-x^2} \psi_2(x) = \int_{-\infty}^{\infty} \varphi_0(x) e^{-x^2} [u_2(x) + a_{20}\varphi_0(x)] dx ,$$

or

$$a_{20} = -\int_{-\infty}^{\infty} u_2(x) e^{-x^2} \varphi_0(x) dx = -\int_{-\infty}^{\infty} x^2 e^{-x^2} \frac{1}{\pi^{1/4}} dx = -\frac{\sqrt{\pi}}{2} \frac{1}{\pi^{1/4}} ,$$

$$a_{21} = -\int_{-\infty}^{\infty} u_2(x) e^{-x^2} \varphi_1(x) dx = -\int_{-\infty}^{\infty} x^2 e^{-x^2} \frac{\sqrt{2}x}{\pi^{1/4}} dx = 0 ,$$

$$\psi_2(x) = x^2 - \frac{\sqrt{\pi}}{2} \frac{1}{\pi^{1/4}} \frac{1}{\pi^{1/4}} = x^2 - \frac{1}{2} .$$

Normalizing,

$$\varphi_2(x) = \frac{\psi_2(x)}{\sqrt{\int_{-\infty}^{\infty} \psi_2(x) e^{-x^2} \psi_2(x) dx}}$$

$$= \frac{x^2 - \frac{1}{2}}{\sqrt{\int_{-\infty}^{\infty} (x^4 - x + \frac{1}{4}) e^{-x^2} dx}}$$

or

$$\varphi_2(x) = \frac{x^2 - \frac{1}{2}}{(\frac{1}{2}\pi^{1/2})^{1/2}} = \frac{2x^2 - 1}{\sqrt{2}\pi^{1/4}}.$$

For $n = 3$, let

$$\psi_3(x) = u_3(x) + a_{30}\varphi_0(x) + a_{31}\varphi_1(x) + a_{32}\varphi_2(x),$$

$$0 = \int_{-\infty}^{\infty} \varphi_0(x) e^{-x^2} \psi_3(x) = \int_{-\infty}^{\infty} \varphi_0(x) e^{-x^2} [u_3(x) + a_{30}\varphi_0(x)] dx,$$

or

$$a_{30} = -\int_{-\infty}^{\infty} u_3(x) e^{-x^2} \varphi_0(x) dx = -\int_{-\infty}^{\infty} x^3 e^{-x^2} \frac{1}{\pi^{1/4}} dx = 0,$$

$$a_{31} = -\int_{-\infty}^{\infty} u_3(x) e^{-x^2} \varphi_1(x) dx = -\int_{-\infty}^{\infty} x^4 e^{-x^2} \frac{\sqrt{2}}{\pi^{1/4}} dx = -\sqrt{2}\frac{3}{4}\pi^{1/4},$$

$$a_{32} = -\int_{-\infty}^{\infty} u_3(x) e^{-x^2} \varphi_2(x) dx = -\int_{-\infty}^{\infty} x^3 e^{-x^2} \frac{x^2 - \frac{1}{2}}{(\frac{1}{2}\pi^{1/2})^{1/2}} dx = 0,$$

$$\psi_3(x) = x^3 - \sqrt{2}\frac{3\pi^{1/4}}{4}\frac{\sqrt{2}x}{\pi^{1/4}} = x^3 - \frac{3}{2}x.$$

Normalizing,

$$\varphi_3(x) = \frac{\psi_3(x)}{\sqrt{\int_{-\infty}^{\infty} \psi_3(x) e^{-x^2} \psi_3(x) dx}}$$

$$= \frac{x^3 - \frac{3}{2}x}{\sqrt{\int_{-\infty}^{\infty} (x^3 - \frac{3}{2}x)^2 e^{-x^2} dx}} = \frac{\pi^{-1/4}}{\sqrt{3}}(2x^3 - 3x)$$

We find that $\varphi_n(x)$ is proportional to the Hermite polynomial. The Hermite polynomials are given by

n	$H_n(x)$
0	1
1	$2x$
2	$-2 + 4x^2$
3	$-12x + 8x^3$
4	$12 - 48x^2 + 16x^4$
5	$120x - 160x^3 + 32x^5$
6	$-120 + 720x^2 - 480x^4 + 64x^6$
7	$-1680x + 3360x^3 - 1344x^5 + 128x^7$
8	$1680 - 13440x^2 + 13440x^4 - 3584x^6 + 256x^8$
9	$30240x - 80640x^3 + 48384x^5 - 9216x^7 + 512x^9$
10	$-30240 + 302400x^2 - 403200x^4 + 161280x^6 - 23040x^8 + 1024x^{10}$

8.3 Independence of $u_n(x) = x^n$; Wronskian

We show the independence of $\{u_n(x) = x^n\}$ using the Wrtonskian determinant.

((**Mathematica**))

Wronskian

```
G[k_] := Table[x^n, {n, 0, k}]

W[k_, m_] := D[G[k], {x, m}]

WR[k_] := Table[W[k, m], {m, 0, k}]

G[6]
```

$\{1, x, x^2, x^3, x^4, x^5, x^6\}$

```
WR[6]
```

$\{\{1, x, x^2, x^3, x^4, x^5, x^6\}, \{0, 1, 2x, 3x^2, 4x^3, 5x^4, 6x^5\}, \{0, 0, 2, 6x, 12x^2, 20x^3, 30x^4\}, \{0, 0, 0, 6, 24x, 60x^2, 120x^3\}, \{0, 0, 0, 0, 24, 120x, 360x^2\}, \{0, 0, 0, 0, 0, 120, 720x\}, \{0, 0, 0, 0, 0, 0, 720\}\}$

```
WR[6] // MatrixForm
```

$$\begin{pmatrix} 1 & x & x^2 & x^3 & x^4 & x^5 & x^6 \\ 0 & 1 & 2x & 3x^2 & 4x^3 & 5x^4 & 6x^5 \\ 0 & 0 & 2 & 6x & 12x^2 & 20x^3 & 30x^4 \\ 0 & 0 & 0 & 6 & 24x & 60x^2 & 120x^3 \\ 0 & 0 & 0 & 0 & 24 & 120x & 360x^2 \\ 0 & 0 & 0 & 0 & 0 & 120 & 720x \\ 0 & 0 & 0 & 0 & 0 & 0 & 720 \end{pmatrix}$$

```
Table[Det[WR[i]], {i, 1, 8}]
```

{1, 2, 12, 288, 34 560, 24 883 200, 125 411 328 000, 5 056 584 744 960 000}

8.4 Gram-Schmidt orthogonalization procedure II

We consider a family of functions

$u_1(x), u_2(x), u_3(x), \ldots\ldots., u_n(x), \ldots.$

where $\{u_n\{x\}\}$ is independent function and

$$(u_i, u_j) = a_{ij} = \int_a^b u_i(x) w(x) u_j(x) dx$$

Suppose that the relation is always valid for the range $a \le x \le b$.

$$a_1 u_1(x) + a_2 u_2(x) + \ldots + a_n u_n(x) = 0$$

Then we have, $a_1 = a_2 = \ldots = a_n = 0$. We now construct a linear combination of the functions

$$\Phi_n(x) = \begin{vmatrix} a_{11} & a_{12} & & a_{1n} \\ a_{21} & a_{22} & & a_{2n} \\ & & & \\ a_{n-1,1} & a_{n-1,2} & & a_{n-1,n} \\ u_1(x) & u_2(x) & & u_n(x) \end{vmatrix}$$

This function is always orthogonal to $u_i(x)$. Thus $\Phi_n(x)$ is orthogonal to

$$\Phi_1(x), \Phi_2(x), ..., \Phi_{n-1}(x).$$

We introduce a Gram determinant, A_n. $A_0 = 1$ and

$$A_n = \begin{vmatrix} a_{11} & a_{12} & & a_{1n} \\ a_{21} & a_{22} & & a_{2n} \\ & & & \\ a_{n-1,1} & a_{n-1,2} & & a_{n-1,n} \\ a_{n,1} & a_{n,2} & & a_{n,n} \end{vmatrix}$$

$$(\Phi_n, \Phi_n) = \begin{vmatrix} a_{11} & a_{12} & & a_{1n} \\ a_{21} & a_{22} & & a_{2n} \\ & & & \\ a_{n-1,1} & a_{n-1,2} & & a_{n-1,n} \\ 0 & 0 & & (u_n, \Phi_n) \end{vmatrix} = (u_n, \Phi_n) A_{n-1}$$

$$(u_n, \Phi_n) = \begin{vmatrix} a_{11} & a_{12} & & a_{1n} \\ a_{21} & a_{22} & & a_{2n} \\ & & & \\ a_{n-1,1} & a_{n-1,2} & & a_{n-1,n} \\ (u_n, u_1) & (u_n, u_2) & & (u_n, u_n) \end{vmatrix} = \begin{vmatrix} a_{11} & a_{12} & & a_{1n} \\ a_{21} & a_{22} & & a_{2n} \\ & & & \\ a_{n-1,1} & a_{n-1,2} & & a_{n-1,n} \\ a_{n,1} & a_{n,2} & & a_{n,n} \end{vmatrix} = A_n$$

Then we have

$$(\Phi_n, \Phi_n) = A_n A_{n-1}$$

or

Then we have the final form of the orthogonal (normalized) functions,

$$\varphi_n = \frac{\Phi_n}{\sqrt{A_{n-1}A_n}} \qquad (n = 1, 2, \ldots)$$

where $A_0 = 1$.

8.5 Legendre polynomials

Gram-Schmidt orthogonalization procedure for the Legendre polynomials. We use the method discussed in **8.4**.

$$w(x) = 1 \text{ and } u_n(x) = x^n, \qquad \text{for } -1 \le x \le 1.$$

((Mathematica))

```
Clear["Global`*"]

weight = 1; u[i_, x_] := x^i;
a[i_, j_] := Integrate[weight u[i, x] u[j, x], {x, -1, 1},
  GenerateConditions → False]

gram[n_] := Det[Table[a[i, j], {i, 0, n}, {j, 0, n}]];

gram[-1] = 1;

Φ[n_, x_] :=
  Det[Append[Table[a[i, j], {i, 0, n - 1}, {j, 0, n}], x^Range[0, n]]] //
   Simplify;

ϕ[n_, x_] := Φ[n, x] / Sqrt[gram[n - 1] gram[n]];

Table[{n, ϕ[n, x], LegendreP[n, x] }, {n, 0, 5}] // Simplify // TableForm
```

0	$\frac{1}{\sqrt{2}}$	1
1	$\sqrt{\frac{3}{2}}\ x$	x
2	$\frac{1}{2}\sqrt{\frac{5}{2}}\ (-1+3\,x^2)$	$\frac{1}{2}\ (-1+3\,x^2)$
3	$\frac{1}{2}\sqrt{\frac{7}{2}}\ x\ (-3+5\,x^2)$	$\frac{1}{2}\ x\ (-3+5\,x^2)$
4	$\frac{3\,(3-30\,x^2+35\,x^4)}{8\sqrt{2}}$	$\frac{1}{8}\ (3-30\,x^2+35\,x^4)$
5	$\frac{1}{8}\sqrt{\frac{11}{2}}\ x\ (15-70\,x^2+63\,x^4)$	$\frac{1}{8}\ x\ (15-70\,x^2+63\,x^4)$



Fig. Normalized Legendre polynomials. n = 0, 1, 2, 3, 4, and 5.

8.6 Hermite polynomials

Gram-Schmidt orthogonalization procedure for the Hermite polynomials.

$w(x) = \exp(-x^2)$ and $u_n(x) = x^n$, for $-\infty \le x \le \infty$.

((Mathematica))

```
Clear["Global`*"]

weight = e^(-x^2); u[i_, x_] := x^i;
a[i_, j_] := Integrate[weight u[i, x] u[j, x], {x, -∞, ∞},
  GenerateConditions → False]

gram[n_] :=  Det[Table[a[i, j], {i, 0, n}, {j, 0, n}]];

gram[-1] = 1;

Φ[n_, x_] :=
  Det[Append[Table[a[i, j], {i, 0, n - 1}, {j, 0, n}], x^Range[0, n]]] //
   Simplify;
```

$$\phi[n\_,\ x\_] := \frac{\Phi[n,\ x]}{\sqrt{\text{gram}[n-1]\ \text{gram}[n]}};$$

```
Table[{n, ϕ[n, x], HermiteH[n, x] }, {n, 0, 5}] // Simplify // TableForm
```

0	$\frac{1}{\pi^{1/4}}$	1
1	$\frac{\sqrt{2}\,x}{\pi^{1/4}}$	$2\,x$
2	$\frac{-1+2\,x^2}{\sqrt{2}\,\pi^{1/4}}$	$-2+4\,x^2$
3	$\frac{x\left(-3+2\,x^2\right)}{\sqrt{3}\,\pi^{1/4}}$	$4\,x\left(-3+2\,x^2\right)$
4	$\frac{3-12\,x^2+4\,x^4}{2\sqrt{6}\,\pi^{1/4}}$	$4\left(3-12\,x^2+4\,x^4\right)$
5	$\frac{x\left(15-20\,x^2+4\,x^4\right)}{2\sqrt{15}\,\pi^{1/4}}$	$8\,x\left(15-20\,x^2+4\,x^4\right)$



Fig. Normalized Hermite polynomials. n = 0, 1, 2, 3, 4, and 5.

8.7 Laguerre polynomials

Gram-Schmidt orthogonalization procedure for the Laguerre polynomials.

$w(x) = \exp(-x)$ and $u_n(x) = x^n$, for $0 \le x \le \infty$.

((Mathematica))

```
Clear["Global`*"]

weight = e^-x; u[i_, x_] := x^i;
a[i_, j_] := Integrate[weight u[i, x] u[j, x], {x, 0, ∞},
  GenerateConditions → False]

gram[n_] := Det[Table[a[i, j], {i, 0, n}, {j, 0, n}]];

gram[-1] = 1;

Φ[n_, x_] :=
  Det[Append[Table[a[i, j], {i, 0, n - 1}, {j, 0, n}], x^Range[0, n]]] //
   Simplify;

ϕ[n_, x_] := Φ[n, x] / Sqrt[gram[n - 1] gram[n]];

Table[{n, ϕ[n, x], LaguerreL[n, x] }, {n, 0, 5}] // Simplify // TableForm
```

0	1	1
1	$-1 + x$	$1 - x$
2	$\frac{1}{2}\left(2 - 4x + x^2\right)$	$\frac{1}{2}\left(2 - 4x + x^2\right)$
3	$\frac{1}{6}\left(-6 + 18x - 9x^2 + x^3\right)$	$\frac{1}{6}\left(6 - 18x + 9x^2 - x^3\right)$
4	$1 - 4x + 3x^2 - \frac{2x^3}{3} + \frac{x^4}{24}$	$1 - 4x + 3x^2 - \frac{2x^3}{3} + \frac{x^4}{24}$
5	$-1 + 5x - 5x^2 + \frac{5x^3}{3} - \frac{5x^4}{24} + \frac{x^5}{120}$	$1 - 5x + 5x^2 - \frac{5x^3}{3} + \frac{5x^4}{24} - \frac{x^5}{120}$

8.8 Mathematica program for the general case

Here we make a Mathematica program called "gramschmidt" for the Gram-Schmidt orthogonalization procedure of more general cases.

General case for Gram-Schmidt procedure

```
Clear["Global`*"]

gramschmidt[weight_, var_, interval_, n_] :=
 Module[{a, gram, Φ, ϕ},
  a[i_, j_] := Integrate[weight var^(i+j) , Join[{var}, interval],
    GenerateConditions → False];
  gram[m_] := Det[Table[a[i, j], {i, 0, m}, {j, 0, m}]]; gram[-1] = 1;
  Φ[m_] :=
   Det[Append[Table[a[i, j], {i, 0, m - 1}, {j, 0, m}], var^Range[0, m]]] //
    Simplify; ϕ[m_] := Φ[m] / Sqrt[gram[m - 1] gram[m]]; Table[{k, ϕ[k] }, {k, 0, n}]]
```

Legendre polynomials

```
gramschmidt[1, x, {-1, 1}, 3]
```

$$\left\{\left\{0, \frac{1}{\sqrt{2}}\right\}, \left\{1, \sqrt{\frac{3}{2}}\, x\right\}, \left\{2, \frac{1}{2}\sqrt{\frac{5}{2}}\left(-1+3x^2\right)\right\}, \left\{3, \frac{1}{2}\sqrt{\frac{7}{2}}\, x\left(-3+5x^2\right)\right\}\right\}$$

Shifted Legendre polynomials

```
gramschmidt[1, x, {0, 1}, 3]
```

$$\left\{\{0, 1\}, \left\{1, 2\sqrt{3}\left(-\frac{1}{2}+x\right)\right\}, \left\{2, \sqrt{5}\left(1-6x+6x^2\right)\right\}, \left\{3, \sqrt{7}\left(-1+12x-30x^2+20x^3\right)\right\}\right\}$$

Hermite polynomials

```
gramschmidt[Exp[-x^2], x, {-∞, ∞}, 3]
```

$$\left\{\left\{0, \frac{1}{\pi^{1/4}}\right\}, \left\{1, \frac{\sqrt{2}\, x}{\pi^{1/4}}\right\}, \left\{2, \frac{-1+2x^2}{\sqrt{2}\,\pi^{1/4}}\right\}, \left\{3, \frac{x\left(-3+2x^2\right)}{\sqrt{3}\,\pi^{1/4}}\right\}\right\}$$

Laguerre polynomials

```
gramschmidt[Exp[-x], x, {0, ∞}, 3]
```

$$\left\{\{0, 1\}, \{1, -1+x\}, \left\{2, \frac{1}{2}\left(2-4x+x^2\right)\right\}, \left\{3, \frac{1}{6}\left(-6+18x-9x^2+x^3\right)\right\}\right\}$$

Associate Laguerre polynomials

```
gramschmidt[x Exp[-x], x, {0, ∞}, 3]
```

$$\left\{\{0, 1\}, \left\{1, \frac{-2+x}{\sqrt{2}}\right\}, \left\{2, \frac{6-6x+x^2}{2\sqrt{3}}\right\}, \left\{3, \frac{1}{12}\left(-24+36x-12x^2+x^3\right)\right\}\right\}$$

Chebyshev I polynomials

$$\texttt{gramschmidt}\left[\frac{1}{\sqrt{1-x^2}}, x, \{-1, 1\}, 3\right]$$

$$\left\{\left\{0, \frac{1}{\sqrt{\pi}}\right\}, \left\{1, \sqrt{\frac{2}{\pi}}\,x\right\}, \left\{2, \sqrt{\frac{2}{\pi}}\left(-1+2x^2\right)\right\}, \left\{3, \sqrt{\frac{2}{\pi}}\,x\left(-3+4x^2\right)\right\}\right\}$$

Shifted Chebyshev I polynomials

$$\texttt{gramschmidt}\left[\frac{1}{\sqrt{x\,(1-x)}}, x, \{0, 1\}, 3\right]$$

$$\left\{\left\{0, \frac{1}{\sqrt{\pi}}\right\}, \left\{1, 2\sqrt{\frac{2}{\pi}}\left(-\frac{1}{2}+x\right)\right\},\right.$$
$$\left.\left\{2, \sqrt{\frac{2}{\pi}}\left(1-8x+8x^2\right)\right\}, \left\{3, \sqrt{\frac{2}{\pi}}\left(-1+18x-48x^2+32x^3\right)\right\}\right\}$$

Chebyshev II polynomials

$$\texttt{gramschmidt}\left[\sqrt{1-x^2}, x, \{-1, 1\}, 3\right]$$

$$\left\{\left\{0, \sqrt{\frac{2}{\pi}}\right\}, \left\{1, 2\sqrt{\frac{2}{\pi}}\,x\right\}, \left\{2, \sqrt{\frac{2}{\pi}}\left(-1+4x^2\right)\right\}, \left\{3, 4\sqrt{\frac{2}{\pi}}\,x\left(-1+2x^2\right)\right\}\right\}$$